```
In[23]:= (* fsr calculations *)

         (* source location and polarisation tensors *)
         (* n = r_hat, l = theta_hat, m = phi_hat *)
         n = {Sin[theta] Cos[phi], Sin[theta] Sin[phi], Cos[theta]};
         l = {Cos[theta] Cos[phi], Cos[theta] Sin[phi], -Sin[theta]};
         m = {-Sin[phi], Cos[phi], 0};

         Print[StringForm["l=``", MatrixForm[l]]];
         Print[StringForm["m=``", MatrixForm[m]]];
         Print[StringForm["n=``", MatrixForm[n]]];

         eP = Outer[Times, l , l] - Outer[Times, m, m];
         eC = Outer[Times, l , m] + Outer[Times, m, l];

         Print[StringForm["eP=``", MatrixForm[eP]]];
         Print[StringForm["eC=``", MatrixForm[eC]]];

         (* unit vectors along detector arms *)
         a = {1, 0, 0};
         b = {0, 1, 0};

         (* dot products of unit vectors along detector arms with polarisation tensors *)
         ePaa = FullSimplify[a.eP.a];
         ePbb = FullSimplify[b.eP.b];
         eCaa = FullSimplify[a.eC.a];
         eCbb = FullSimplify[b.eC.b];
         (*
          Print[StringForm["ePaa=``",ePaa]];
          Print[StringForm["ePbb=``",ePbb]];
          Print[StringForm["eCaa=``",eCaa]];
          Print[StringForm["eCbb=``",eCbb]];
          *)

         (* LW antenna patterns as sanity check *)
         Fp = 0.5 * (ePaa - ePbb);
         Fc = 0.5 * (eCaa - eCbb);
         Print[StringForm["Fp=``", Fp]];
         Print[StringForm["Fc=``", Fc]];

         (* compare with standard expressions in the literature *)
         Simplify[Fp - (1 / 2) Cos[2 phi] (1 + Cos[theta] ^ 2)]
         Simplify[Fc - (-1) Sin[2 phi] Cos[theta]]
```

$$l=\begin{pmatrix} \text{Cos[phi] Cos[theta]} \\ \text{Cos[theta] Sin[phi]} \\ \text{-Sin[theta]} \end{pmatrix}$$

$$m=\begin{pmatrix} \text{-Sin[phi]} \\ \text{Cos[phi]} \\ 0 \end{pmatrix}$$

$$n=\begin{pmatrix} \text{Cos[phi] Sin[theta]} \\ \text{Sin[phi] Sin[theta]} \\ \text{Cos[theta]} \end{pmatrix}$$

```
ePlus=( Cos[phi]^2 Cos[theta]^2 - Sin[phi]^2                 Cos[phi] Sin[phi] + Cos[phi] Cos[theta]^2 Sin[phi]
        Cos[phi] Sin[phi] + Cos[phi] Cos[theta]^2 Sin[phi]    -Cos[phi]^2 + Cos[theta]^2 Sin[phi]^2
        -Cos[phi] Cos[theta] Sin[theta]                       -Cos[theta] Sin[phi] Sin[theta]
```

```
eC=( -2 Cos[phi] Cos[theta] Sin[phi]                   Cos[phi]^2 Cos[theta] - Cos[theta] Sin[phi]^2   Sin[phi]
     Cos[phi]^2 Cos[theta] - Cos[theta] Sin[phi]^2     2 Cos[phi] Cos[theta] Sin[phi]                  -Cos[phi
     Sin[phi] Sin[theta]                               -Cos[phi] Sin[theta]                            0
```

```
Fp=0.5 (Cos[phi]^2 + Cos[phi]^2 Cos[theta]^2 - Sin[phi]^2 - Cos[theta]^2 Sin[phi]^2)

Fc=0.5 (-2 Cos[phi] Cos[theta] Sin[phi] - Cos[theta] Sin[2 phi])
```

Out[43]= 0.

Out[44]= 0.

In[45]:=

```
(* exact antenna pattern functions at fsr *)

(* arm lengths for H1, H2 in meters and seconds *)
L1 = 4000;
L2 = 2000;
c = 300000000;
T1 = L1 / c;
T2 = L2 / c;

(* fsr for H1 *)
f = 1 / (2 T1);
Print[StringForm["fsr = ``", f]];

(* response functions *)
D1a = (1 / 2) * Exp[-I 2 Pi f T1] *
   (Exp[I Pi f T1 (1 + a.n)] Sin[Pi f T1 (1 - a.n)] / (Pi f T1 (1 - a.n)) +
     Exp[-I Pi f T1 (1 - a.n)] Sin[Pi f T1 (1 + a.n)] / (Pi f T1 (1 + a.n)));

D1b = (1 / 2) * Exp[-I 2 Pi f T1] *
   (Exp[I Pi f T1 (1 + b.n)] Sin[Pi f T1 (1 - b.n)] / (Pi f T1 (1 - b.n)) +
     Exp[-I Pi f T1 (1 - b.n)] Sin[Pi f T1 (1 + b.n)] / (Pi f T1 (1 + b.n)));

D2a = (1 / 2) * Exp[-I 2 Pi f T2] *
   (Exp[I Pi f T2 (1 + a.n)] Sin[Pi f T2 (1 - a.n)] / (Pi f T2 (1 - a.n)) +
     Exp[-I Pi f T2 (1 - a.n)] Sin[Pi f T2 (1 + a.n)] / (Pi f T2 (1 + a.n)));

D2b = (1 / 2) * Exp[-I 2 Pi f T2] *
   (Exp[I Pi f T2 (1 + b.n)] Sin[Pi f T2 (1 - b.n)] / (Pi f T2 (1 - b.n)) +
     Exp[-I Pi f T2 (1 - b.n)] Sin[Pi f T2 (1 + b.n)] / (Pi f T2 (1 + b.n)));

G1p = (1 / 2) * (ePaa D1a - ePbb D1b);
G1c = (1 / 2) * (eCaa D1a - eCbb D1b);
G2p = (1 / 2) * (ePaa D2a - ePbb D2b);
G2c = (1 / 2) * (eCaa D2a - eCbb D2b);
```

fsr = 37500

```
In[60]:=
  (* overlap reduction function for H1-H2 and acceptance functions D1, D2 at fsr*)

  orf12int = (5 / (8 * Pi)) * Sin[theta] * (G1p Conjugate[G2p] + G1c Conjugate[G2c]);
  orf11int = (5 / (8 * Pi)) * Sin[theta] * (G1p Conjugate[G1p] + G1c Conjugate[G1c]);
  orf22int = (5 / (8 * Pi)) * Sin[theta] * (G2p Conjugate[G2p] + G2c Conjugate[G2c]);

  (* split up integrals into two parts since denominator of integrand → 0 at theta=
    Pi/2 *)
  orf12 = NIntegrate[orf12int, {theta, 0, Pi / 2}, {phi, 0, 2 * Pi}] +
     NIntegrate[orf12int, {theta, Pi / 2, Pi}, {phi, 0, 2 * Pi}];
  orf11 = NIntegrate[orf11int, {theta, 0, Pi / 2}, {phi, 0, 2 * Pi}] +
     NIntegrate[orf11int, {theta, Pi / 2, Pi}, {phi, 0, 2 * Pi}];
  orf22 = NIntegrate[orf22int, {theta, 0, Pi / 2}, {phi, 0, 2 * Pi}] +
     NIntegrate[orf22int, {theta, Pi / 2, Pi}, {phi, 0, 2 * Pi}];

  (* take real part, since very small imaginary part due to machine round-off *)
  D1 = Sqrt[Re[orf11]];
  D2 = Sqrt[Re[orf22]];

  Print[StringForm["orf12 = ``", orf12]];
  Print[StringForm["D1 = ``", D1]];
  Print[StringForm["D2 = ``", D2]];
```

orf12 = $4.80273 \times 10^{-12} + 0.0437255\, i$

D1 = 0.18525258845665613`

D2 = 0.615536504280289`